
POLYNOMIAL MULTIPLICATION

PURPOSE
Carry out the multiplication of 2 polynomials with real coefficients.

SYNTAX
LET <v3> = POLYNOMIAL MULTIPLICATION <v1> <v2> <SUBSET/EXCEPT/FOR qualification>
where <v1> is the variable whose elements are the ordered real coefficients of the first polynomial;

 <v2> is the variable whose elements are the ordered real coefficients of the second polynomial;
 <v3> is the variable whose elements are the ordered real coefficients of the resultant polynomial;

and where the <SUBSET/EXCEPT/FOR qualification> is optional and rarely used in this context.

EXAMPLES
LET Y3 = POLYNOMIAL MULTIPLICATION Y1 Y2

NOTE
The first element of the variable is the coefficient of the constant term, the second element is the coefficient of the linear term, the third
element is the coefficient of the quadratic term, the fourth element is the coefficient of the cubic term, and so on. Thus the polynomial 4
+ 11*X + 37*X2 + 8*X3 + 19*X4 can be stored in the variable Y with the following command:

LET Y = DATA 4 11 37 8 19

DEFAULT
None

SYNONYMS
None

RELATED COMMANDS
LET = Evaluates general functions.
POLYNOMIAL ADDITION = Carries out a polynomial addition.
POLYNOMIAL SUBTRACTION = Carries out a polynomial subtraction.
POLYNOMIAL DIVISION = Carries out a polynomial division.
POLYNOMIAL SQUARE = Carries out a polynomial square.
POLYNOMIAL EVALUATION = Carries out a polynomial evaluation.
COMPLEX ROOTS = Computes the roots of a complex polynomial.
PLOT = Plots data or functions
COMPLEX MULTLIPLICATION = Carries out a complex multiplication.
VECTOR DOT PRODUCT = Computes the vector dot product.
MATRIX MULTIPLICATION = Carries out a matrix multiplication.

APPLICATIONS
Mathematics

IMPLEMENTATION DATE
87/10

PROGRAM
. PURPOSE--DETERMINE THE COMPLEX ROOTS OF A SUM OF 20 QUINTICS
.          WHERE THE K-TH QUINTIC IS OF THE FORM (1 + (1/K)*X)**5
. ANALYSIS TECHNIQUE--POLYNOMIAL ARITHMETIC + LAGUERRE’S METHOD FOR
.                     FINDING COMPLEX ROOTS OF COMPLEX POLYNOMIALS
.                     (ALTHOUGH IN THIS CASE, THE POLYNOMIAL IS REAL).
. APPLICATION--FILTER DESIGN, PROBABILITY THEORY
. NOTE--FOR TESTING PURPOSES--
.       THE RESULTING POLYNOMIAL HAS COEFFICIENTS--
.             0.2000000E+02
.             0.1798870E+02
.             0.1596163E+02
.             0.1200868E+02
.             0.5411423E+01
.             0.1036926E+01
.       WITH COMPLEX ROOTS--
.            -0.2387324E+01  0.0000000E+00 (DOUBLE ROOT)
.            -0.1639799E+01 -0.1343364E+01
.            -0.1639799E+01  0.1343364E+01
.             0.2241036E+00 -0.1322019E+01
.             0.2241036E+00  0.1322019E+01
FEEDBACK OFF; DIMENSION 100 VARIABLES
LET N = 20
.      STEP 1--DEFINE THE 20 MONOMIALS
LOOP FOR K = 1 1 N

LET M^K (1) = 1
LET M^K (2) = 1/K

END OF LOOP
.      STEP 2--DEFINE THE 20 POLYNOMIALS
LOOP FOR K = 1 1 N

LET P^K = POLYNOMIAL MULTIPLICATION M^K M^K
LET P^K = POLYNOMIAL MULTIPLICATION P^K M^K
LET P^K = POLYNOMIAL MULTIPLICATION P^K M^K
LET P^K = POLYNOMIAL MULTIPLICATION P^K M^K

END OF LOOP
.      STEP 3--COMPUTE THE SUM OF THE 20 POLYNOMIALS
LET PSUM = DATA 0 0 0 0
LOOP FOR K = 1 1 N

 LET PSUM = POLYNOMIAL ADDITION PSUM P^K
END OF LOOP
.      STEP 4--PLOT THE FUNCTION
MULTIPLOT 2 1; MULTIPLOT CORNER COORDINATES 0 0 100 100
LET X = SEQUENCE -5 .1 5
LET Y = POLYNOMIAL EVALUATION PSUM X
TITLE SUM OF POLYNOMIALS
PLOT Y X
.      STEP 5--FIND THE COMPLEX ROOTS
LET X2 Y2 = COMPLEX ROOTS PSUM
.      STEP 6--PLOT THE COMPLEX ROOTS
CHAR X
LINES
GRID ON
X1LABEL REAL COMPONENT
Y1LABEL COMPLEX COMPONENT
TITLE PLOT ROOTS
PLOT Y2 X2
END OF MULTIPLOT

